

小神童趣味电脑系列

沈 漥

张云辉

编著

格 诚

电脑益智游戏



益智游戏 寓教于乐
愉快学习电脑

辽宁科学技术出版社

8.2
3

丛书前言

电脑在我们生活的世界里,几乎是无处不有,无处不用。掌握电脑技术已成为现代人的重要标志。有远见卓识的家长已经或正准备为自己的孩子购置电脑,以使孩子有能力迎接新世纪的挑战。但是,不容忽视的是,在一些家庭里,孩子只是用电脑玩一些无益的游戏,更有甚者涉足黄毒。这个问题已引起了包括专家、家长在内的全社会的重视。针对这一情况,我们组织了一批有经验的电脑普及教育专家,编写了这套系列丛书《小神童趣味电脑系列》。本系列丛书选择了一系列既有知识性,又有趣味的题目,编制了一些简短的程序。我们试图让孩子们通过输入并运行这些有趣的程序,能够熟悉电脑操作,了解程序的编写方法思路,同时学习一些相关知识。

目前,中小学电脑教育普及率已达到了相当的水平。中小学电脑教学已从操作逐步转向编程。针对这一发展趋势,本系列丛书对每个题目的程序都做了分析,对关键程序语句做了特别讲解。因此,本书也可作为中小学电脑教学的辅助教材。

本系列丛书荣幸地聘请到了中国计算机学会普及委员会主任、国际信息学奥林匹克中国队总教练、清华大学教授吴文虎先生任顾问。吴文虎教授对本系列丛书的编写给予了大力支持,在此我们表示衷心感谢。

本系列丛书第一辑奉献给小读者的有:《电脑益智游戏》、《电脑音乐》、《LOGO 绘图》和《电脑趣味数学》。

编者

1996 年 9 月



电脑游戏的趣味性很强，那么什么样的电脑游戏既有趣味性又有益开发智力、帮助学习呢？这需要有选择、有目的地进行。这本《电脑益智游戏》正是从这个角度出发，选编了一些益智电脑游戏，希望能达到寓教于乐的效果。

趣味游戏主要包括以下几方面内容：

(1) 数字类。有许多奇妙的数字游戏是同学们非常感兴趣的，例如把某些数排成一定的阵势。所以我们选编了“魔方阵”、“解幻方”等游戏。

(2) 智力类。有很多开发智力的游戏非常适合用程序来完成，像我们选编的“破案”、“聪明的酒鬼”、“渡河”、“巧分糖果”等。

(3) 娱乐类。学习之余，放松一下，不妨进行这类游戏，像“猜字母”、“猜数游戏”、“23根火柴”等。

(4) 棋类。进行棋类活动既可消遣，又能开发智力，像“五子连珠”、“斜线跳棋”、“十五子棋”等。

(5) 概率类。如“用‘降雨法’求 π 值”、“珠子在钉板上下落的游戏”等。

(6) 画图类。如“希望号飞艇”、“天涯孤旅”等。

(7) 动画类。在屏幕上欣赏动画一定会使同学们既新奇又满足，像“打台球”、“奔驰汽车”等。

(8) 英语测试类。如“英语不规则动词的测试”、“说出同义词”等。

(9) 观赏类。如“迷宫里的老鼠”、“生命的演变”、“电子钟”等。

(10) 其他类。如“可供查询的万年历”、“破译

密码”等。

书中的所有程序都在 80286 机上用 BASICA 调试通过。读者可在各种 PC 机上运行本书所有程序。

作者

1996 年 10 月



丛书前言

前言

1. 聪明的求婚者	1
2. 宴会排座	4
3. 红白黑球的取法	7
4. 破案	10
5. 聪明的酒鬼	14
6. 比高矮	18
7. 猜字母	23
8. 巧取糖果	28
9. 八皇后问题	33
10. 渡河	38
11. 狐狸追杀兔子	42
12. 捉蛀虫	47
13. 交友游戏	53
14. 猜数游戏	57
15. 找零钱	61
16. 23根火柴	67
17. 石头、剪子、布	71
18. 用“降雨法”求 π 值	77
19. 珠子在钉板上下落的游戏（高登板）	81
20. 掷骰子，算点数	85
21. 人口预测	90
22. 生命的演变	94
23. 转圈的数字	101
24. 魔方阵	105

25. 解幻方	109
26. 英语不规则动词的测验	114
27. 说出同义词	119
28. 移子游戏	125
29. 五子连珠	130
30. 十五子棋	136
31. 斜线跳棋	143
32. 破译密码	150
33. 难题	156
34. 哥尼斯堡桥问题	162
35. 带倒影的火车头	168
36. 希望号飞艇	171
37. 带音乐的贺卡	176
38. 打台球	180
39. 电子钟	185
40. 奔驰汽车	189
41. 天涯孤旅	193
42. 保龄球	198
43. 迷宫里的老鼠	205
44. 移梵塔	212
45. 可供查询的万年历	221
参考文献	230

1. 聪明的求婚者

一、游戏介绍

英国大文豪莎士比亚在其名著《威尼斯商人》中描绘了这样一段情节：女主角鲍西娅对求婚者说：“这里有三个盒子，一个是金盒子，一个是银盒子，一个是铅盒子，每个盒子上的铭牌各写有一句话。三句话中，只有一句是真话，谁能猜中我的肖像放在哪一个盒子里，谁就能做我的丈夫。”盒子及铭牌上的话如图 1 所示。

肖像在这盒里 肖像不在这盒里 肖像不在金盒里




图 1

那么，你知道鲍西娅的未婚夫是怎样猜的吗？你不想也试一下吗？

二、编程思路

用 J、Y、Q 这三个变量代表金、银、铅三个盒子，其值皆可取 0 或 1，取 0 表示里面没有肖像，取 1 表示装有肖像。

用 A、B、C 这三个变量表示铭牌上的话，A 为金盒子上铭牌上的话，B、C 则分别为银盒子、铅盒子上的话，其值为 0 或 1，表示是谎话或真话。然后用三重循环来排出装或未装肖像的所有情况（即每个盒子都可能装或未装），并逐一加以判断，如果满足：（1）A、B、C 的值只有其中一个为 1，其他两个全为 0；（2）J、Y、Q 的值中只有一个为 1，其他两个全为 0。那么，值为 1 的那个变量所代表的盒子里装有肖像，其他两个盒子当然是空的（变量的值为 0）。注意，条件（2）是一个较隐蔽的条件，是从题意中分析得出的。

三、游戏程序

```
10 FOR J=0 TO 1
20 FOR Y=0 TO 1
30 FOR Q=0 TO 1
40 IF J=1 THEN A=1 ELSE A=0
50 IF Y=0 THEN B=1 ELSE B=0
60 IF J=0 THEN C=1 ELSE C=0
70 IF J+Y+Q<>1 GOTO 130
80 IF A+B+C<>1 GOTO 130
90 PRINT "J", "Y", "Q"
100 PRTNT J, Y, Q: PRINT
110 PRINT "A", "B", "C"
120 PRINT A, B, C
130 NEXT Q
140 NEXT Y
150 NEXT J
160 END
```

四、运行结果

RUN

J	Y	Q
0	1	0 (银盒子里有肖像)

A	B	C
0	0	1 (铅盒子铭牌上写的是真话)

五、程序说明

40~60 语句：根据盒子上铭牌上的话确定其逻辑关系；

70 语句：只有一个盒子里有肖像；

80 语句：只有一句话是真；

90~120 语句：输出结果。

2. 宴会排座

一、游戏介绍

在一次宴会上，有八个来自不同国家的宾客被安排围着一张圆桌就座。A 是中国人，会讲英语；B 是意大利人，会讲西班牙语；C 是英国人，会讲法语；D 是日本人，会讲汉语；E 是法国人，会讲德语；F 是俄国人，会讲意大利语；G 是西班牙人，会讲日语；最后一位是德国人，会讲俄语。现在请你想办法安排好他们各自的座位，使他们能在宴会上愉快地跟两旁的客人交谈。

二、编程思路

要想使每个人都能自如地跟左右两旁的客人交谈确实不太容易，我们可以这样做：因为每位外宾除了母语外，还都会另一门外语，为方便起见，把所有客人会的语言编号如下：

语言种类：汉 英 意 西 法 日 德 俄

代 号： 1 2 3 4 5 6 7 8

把每个人的代号(A~H)跟他会的两种语言代号连起来放入字符串数组 P\$ 的每个元素内。例如 A12 和 B34 分别表示：中国人会讲汉语与英语，意大利人会讲意大利语和西班牙语。

把 A 安排在某一座位，设为第一号位置，然后在其余的 7 位客人中寻找会讲英语的客人，即 P\$ (1) 的最末字符是 2，从 P\$ (2) 到 P\$ (7) 中找出第二个字符是 2 的，找到后立即放在第二号位置，原字符串替换成“L”，表示已安排，重复以上过程，即可妥善地排好座位。

三、游戏程序

```
10 PRINT: PRINT "ZUO WEI HAO MA:";
20 FOR I=1 TO 8
30 PRINT I;
40 READ P$ (I);
50 NEXT I
60 PRINT: PRINT: PRINT "KE REN DAI
HAO; A";
70 X$ =MID$ (P$ (1), 3, 1)
80 FOR I=2 TO 8
90 FOR J=2 TO 8
100 IF MID$ (P$ (J), 2, 1) <>X$ GOTO
    150
110 PRINT MID$ (P$ (J), 1, 1); " ";
120 X$ =MID$ (P$ (J), 3, 1)
130 P$ (J) = "L"
140 NEXT J
150 NEXT I
170 DATA A12, B34, C25, D61, E57, F83, G46,
    H78
180 END
```

四、运行结果

RUN

ZUO WEI HAO MA: 1 2 3 4 5 6 7 8
KE REN DAI HAO:A C E H F B G D

五、程序说明

10 语句：打印座位号码；

20~40 语句：读入数组 P\$ 各元素；

50 语句：打印“客人代号：”及第一位客人代号 A；

60 语句：找出 P\$ (1) 第三个字符并赋给字符串变量 X\$；

70~140 语句：二重循环把其余元素的第二个字符找出来并与 X\$ 比较，若相等，由 100 语句打印第一字符，即客人代号，并把这个元素的第三个字符重新赋给 X\$，由 120 语句换为“L”，表示已安排，下次不再访问。

把 A 安排在第一号位置，其他客人依号入座，当然第一号与第八号相邻。

如果有更多的客人需要排座，可将本程序进行扩充。具体方法请读者自行考虑。

3. 红白黑球的取法

一、游戏介绍

假设口袋里放着 12 个球，其中 3 个是红的，3 个是白的，6 个是黑的，现从中任取 8 个球，那么能有多少种不同的取法呢？能给出具体的取法吗？如果有条件，可动手做一做，验证一下答案。

二、编程思路

编程时，考虑到取出的 8 个球中红球可能的个数为 0, 1, 2, 3。白球可能的个数为 0, 1, 2, 3。而黑球可能的个数为 2, 3, 4, 5, 6。因此可利用二重循环，A 为外循环控制变量，从 0 变到 3，B 为内循环控制变量，从 0 变到 3。则 $C=8-A-B$ 。若 $A+B \geq 2$ ，就求出 C，并打印结果，否则继续运算，直至循环结束，就可得到全部取法。

程序中，A 代表红球，B 代表白球，C 代表黑球。

三、游戏程序

```
10 D=0  
20 FOR A=0 TO 3  
30 FOR B=0 TO 3
```

```
40 IF A+B<2 GOTO 120
50 C=8-A-B
60 D=D+1
70 PRINT "No." D
80 IF A<>0 THEN FOR I=1 TO A:
    PRINT "a";:NEXT I
90 IF B<>0 THEN FOR I=1 TO B:
    PRINT "b";:NEXT I
100 FOR I=1 TO C: PRINT "c";:NEXT I
110 PRINT
120 NEXI B
130 NEXT A
140 PRINT "d="; D
150 END
```

四、运行结果

RUN

No. 1	bbccccc
No. 2	bbbcccc
No. 3	abcffff
No. 4	abbcccc
No. 5	abbbccc
No. 6	aacccccc
No. 7	aabcccc
No. 8	aabbcccc
No. 9	aabbccc
No. 10	aaacccccc

No. 11 aaabcccc

No. 12 aaabbccc

No. 13 aaabbbcc

d=13

五、程序说明

20、130 语句：按红球的各种可能性分别进行验算；

30、120 语句：按白球的各种可能性分别进行验算；

50 语句：计算黑球数目；

60 语句：统计取法数目；

80~100 语句：打印各种取法；

140 语句：打印总的可能数。

4. 破案

一、游戏介绍

公安局长接到一项紧急任务，要他在代号为 A、B、C、D、E、F 六个队员中挑选出若干人去侦破一件案子，人选的配备要求，必须注意以下各点：

- (1) A、B 二人中至少去一人；
- (2) A、D 不能一起去；
- (3) A、E、F 三人中要派两人去；
- (4) B、C 两人中都去或都不去；
- (5) C、D 两人中去一人；
- (6) 若 D 不去，则 E 也不去。

那么，局长最后到底都派谁了呢？

二、编程思路

建立数组 N，将 1~6 送入数组 N 中。1~6 表示 A~F。然后利用四重循环，将数组 N 中的元素每四个组合在一起。每一次组合后的结果送入数组 M 中去，然后考查数组 M 中的数是否满足题目要求的 6 个条件。如果不满足，取另一种组合送入 M，重新考查。如果满足条件，打印出数组 M，即是所求答案。

三、游戏程序

```
10 DIM N (6), M (4), A$ (20)
20 N=1
30 FOR P=1 TO 6
40 N (P) =P
50 NEXT P
60 FOR I=1 TO 3
70 M (1) =N (I)
80 FOR J=I+1 TO 4
90 M (2) =N (J)
100 FOR K=J+1 TO 5
110 M (3) =N (K)
120 FOR L=K+1 TO 6
130 M (4) =N (L)
140 A$ (N) =STR$ (M (1)) +STR$ (M (2))
    +STR$ (M (3)) +STR$ (M (4))
150 N=N+1
160 NEXT L
170 NEXT K
180 NEXT J
190 NEXT I
200 A1$ = "1"
210 A2$ = "2"
220 A3$ = "3"
230 A4$ = "4"
240 A5$ = "5"
```